

Designer power-supplies

Walter Gray's linear power supply designer is simple and written in Qbasic. Nevertheless, it takes into account transformer winding resistance and produces a comprehensive list of ratings.

Most electronics textbooks manage to discuss the basic mains transformer-rectifier-capacitor power supply without actually telling you how to calculate the output voltage. If you want to calculate the psu output before building it then the methods available are either graphical, such as the one referred to at the end of this article, or a simulation package, assuming you can find all the required model parameters.

My short computer program, described here, allows you to calculate pretty well everything you will need for a practical power supply design. It is written in Qbasic, making it accessible to anyone with a reasonably modern pc.

Structurally, the program is extremely simple, and it will be easy to convert it to another language. I have even shoehorned it into a Casio pocket computer, though the run time was about 30 seconds. On a 50MHz 486 pc however the run-time is a fraction of a second.

How the program works

The core of the program is a calculation of the

charging and discharging curves of the reservoir capacitor, **Fig. 1**. The charging period from time t_1 to t_2 is described by a cumbersome expression. This expression is the solution of the differential equation for charging from a sinusoidal source, **Fig. 2**. Also shown in this diagram is the discharge equation, for the period t_2 to t_1 .

Initially, the program assumes that the capacitor is fully charged. It then iteratively discharges and recharges it until the ripple voltage reaches a steady value. Times t_1 and t_2 are re-estimated at each iteration.

This is a crude procedure, but it is justifiable as the iteration seems to be strongly convergent. The convergence error and computed voltages are printed by the program, so you can check that it is producing reasonable answers.

Peak ripple and rectifier currents are estimated by assuming the capacitor is at its mean voltage and is charged from the peak supply voltage. This sometimes gives an overestimate, but agrees with the practical observation that peak ripple current is essentially independent of the capacitor value for reasonable designs.

The main simplifying assumption in this program is that the rectifiers are treated as ideal switches. In practice they actually turn on and off quite slowly, as can be seen with an oscilloscope. However, this does not appear to be a major source of error. Equivalent series resistance of the capacitor is also ignored. This tends to increase the ripple crest voltage slightly, but for most practical designs the effect is not important.

How to use the program

The first step is to specify the load current and/or the load resistance. Nowadays the majority of new designs will include a voltage regulator, so an unvarying load will appear as a constant current load.

You can specify a load current by setting




Fig. 1. Voltage waveforms typical of a linear power supply.

resistance to 1MΩ, or you can specify load resistance by setting current to zero. Alternatively, a combination of the two can be used.

For simplicity, a constant current load is assumed to be compliant right down to zero volts, though in practice you will have to specify a minimum value for the ripple trough voltage.

You next need to measure three parameters for the mains transformer. These parameters are listed in comments at the head of the program and are measured off-load. For all but a high current supply, a digital multimeter will suffice for this job. Where high currents are involved, a low-ohm meter with four-terminal measurement will be needed to check the secondary winding resistance. For a centre-tapped secondary, use the averages of the two secondary resistances and turns-ratios.

These three parameters *must* be measured as the manufacturer's figures are often approximate or non-existent. If you do not yet have

Table. Values calculated using the psu design program versus measured values demonstrate the usefulness of the program.

	Calculated	Measured
Mean capacitor voltage	37.35295V	37.4V
Ripple voltage	1.301182V pk-pk	1.28V pk-pk
Peak ripple current	3.340345A	3.4A

the transformer to hand, you can use the program to estimate the allowable limits for winding resistances.

There may be other resistances to budget for too. A fuse-holder in the secondary circuit for example may add as much as 0.2Ω, which should be included as part of the transformer secondary resistance.

Figures of merit

The next step is to choose a value for the reservoir capacitor. Schade used a dimensionless figure-of-merit. This is $2\pi fCR$, where

f is the mains frequency, C is the capacitance and R is the equivalent load resistance ($R=V_{op}/I_{op}$).

Capacitance C is chosen so that the figure-of-merit is around 10-15 for full-wave rectification or 20-30 for half-wave. The figure is not critical, so there is no need to go hunting for strange-valued capacitors. In any case the tolerances on electrolytic capacitors are very loose and the actual value has very little effect on the mean output voltage. If the capacitance is too low then the ripple voltage increases disproportionately and regulation will suffer. There is little to be gained by setting it too high.

The figure-of-merit is equivalent to saying that the time constant CR should be about five charging cycles. Easier still, you can try a few different values and let the computer do the work for you.

Finally, use 'wave' and Nrec to define the circuit topology. Wave is 0.5 for a half-wave circuit and 1.0 for full-wave. Variable Nrec will be 1 for half-wave or centre-tapped full-wave circuits and 2 for full-wave bridge circuits. Vrec should be acceptable if left at 0.7 V, though modern designs may well use Schottky rectifiers where Vrec will be about 0.5 V.

Output values printed at the end of the run




Fig. 2. Circuit model for charging and discharging of the linear supply's smoothing capacitor.

Charging equation

$$C \frac{dV}{dt} + \frac{V}{R} + I_c = \frac{V_p \cos(2\pi f(t - T_c)) - V - V_d}{R_s}$$

Discharging equation

$$C \frac{dV}{dt} + \frac{V}{R} + I_c = 0$$

Qbasic listing for designing power supplies. Known data is entered on the DATA line. Values currently following the DATA statement are for the example mentioned in the text.

```

REM Program psu6.bas for mains power-supply calculations.
REM For transformer-rectifier-capacitor circuits with sine-wave
REM mains supply and load specified as current and/or resistance.

REM By W. Gray, started 4-Aug-95, this version 6-Jan-97

DECLARE FUNCTION arcos (A)

DATA 1, 1e6, 237.3, 50, .1354,33.3, 0.88, 5000e-6,0.7, 1, 2
READ Ic, Rload, Vsup, freq, Trat, Rpri, Rsec, cap, Vrec, wave, Nrec

REM load group
REM Ic, load current, A
REM Rload,load resistance, ohms

REM mains supply group
REM Vsup, mains supply to transformer primary, volts rms
REM freq, supply frequency, Hz

REM components group
REM Trat, transformer turns ratio, secondary/primary
REM Rpri, transformer primary resistance, ohms
REM Rsec, transformer secondary resistance, ohms
    
```

```

REM cap, reservoir capacitor, farads
REM Vrec, forward voltage drop of single rectifier

REM configuration group
REM wave, 0.5=half-wave, 1.0=full wave rectification
REM Nrec, total number of rectifiers in series

CLS : REM initialise
Vsec = Vsup * Trat
Vpeak = Vsec * SQR(2)
Rsec = Rsec + Rpri * Trat * Trat
REM approximate allowance for dynamic resistance of rectifiers
Iop = Ic + Vsec / Rload: Rsou = Rsec + .025 * Nrec / Iop
Vrec = Nrec * Vrec
Tcyc = .5 / wave / freq
pi2f = freq * 6.2831853#
rc = cap * Rload
cr = cap * Rsou
A = (1 + Rsou / Rload) / cr
b = Vpeak / cr
c = b / (A * A + pi2f * pi2f)
d = (Vrec + Ic * Rsou) / cr
e = d / A
f = Ic * Rload
t1 = Tcyc
t2 = 0
V2 = Vpeak
    
```

are mostly self-explanatory. Various peak currents are needed to specify the capacitor, rectifiers and fuses. When using a regulator, the crest and trough voltages have to be known so that the regulator limits are not exceeded. These printouts only solve part of the overall psu design problem. However, you should need no further assistance to derive power dissipation, regulation factor, rectifier peak-inverse voltage, etc.

You can also vary data values to explore sensitivity to component tolerances and worst-case load and mains variations or to adjust a design to meet particular constraints.

The data statement at the head of the program lists values for a circuit that I constructed and measured. The exact calculated results are listed in the Table, so you can check that the program is working correctly. Measured values are also listed so you can see the sort of errors you might expect. The circuit in question had a high figure-of-merit. In general the calculated values become less accurate as figure-of-merit reduces.

Conclusion

Now you can throw away those tattered old photocopies of Schade's graphs. The program described here should give all that is needed to design the basic mains power supply, though at the minor cost of having to do some simple component measurements. To go further it would be necessary to use a simulation package, though the accuracy of the results at ordinary mains frequencies would probably not be much better.

Reference

Schade, O H, 'Analysis of rectifier operation', Proc IRE, 1943, Vol. 31, No 7.

Output from the power supply design program. Note that formatting of the data has been edited slightly to increase clarity.

n	Error	Vop	Vrip
1	45439.35	43.40221	-7.395554E-02
2	.1363602	41.60973	6.240463E-02
3	.3032188	40.24191	.3656235
4	.2762871	39.2697	.6419106
5	.2124672	38.60608	.8543777
6	.1509972	38.16445	1.005375
7	.102787	37.87523	1.108162
8	6.821442E-02	37.68774	1.176376
9	4.458618E-02	37.56697	1.220963
10	2.885818E-02	37.4895	1.249821
11	1.856995E-02	37.43993	1.268391
12	1.190567E-02	37.40827	1.280296
13	7.614136E-03	37.38807	1.28791
14	4.859924E-03	37.37518	1.29277
15	3.105164E-03	37.36697	1.295876
16	1.979828E-03	37.36174	1.297855
17	1.25885E-03	37.35841	1.299114
18	8.010864E-04	37.35628	1.299915
19	5.149841E-04	37.35493	1.30043
20	3.242493E-04	37.35406	1.300755
21	2.098083E-04	37.35352	1.300964
22	1.296997E-04	37.35317	1.301094
23	8.773804E-05	37.35295	1.301182

constant current load= 1 A
 resistive load= 1000000 ohms
 supply voltage= 237.3 Vrms at 50 Hz
 peak secondary voltage= 45.43928 V
 effective source resistance= 1.540493 ohms
 reservoir capacitor= 5000 microfarads

mean output= 37.35295 V, current= 1.000037 A
 ripple= 1.301182 VPTP
 ripple crest= 38.00354 V, trough= 36.70236 V
 peak capacitor ripple current= 3.340345 A
 peak repetitive rectifier current= 4.340383 A
 peak non-repetitive (inrush) current= 28.58779 A
 approx inrush duration= 7.702462 milliseconds
 psu figure-of-merit= 58.67169

```
Vprev = 1000 * Vpeak
Error = Vprev
n = 0

REM main loop
DO WHILE Error > .0001 AND n < 200
n = n + 1

REM discharge capacitor from t2 to t1
V1 = (V2 + f) * EXP((t2 - t1) / rc) - f
t1 = Tcyc - arcos((V1 + Vrec) / Vpeak) / pi2f

REM charge capacitor from t1 to t2
temp = A * COS(pi2f * (t1 - Tcyc)) + pi2f *
SIN(pi2f * (t1 - Tcyc))
Q = V1 + e - c * temp
Q = Q * EXP(A * (t1 - t2 - Tcyc))
temp = A * COS(pi2f * t2) + pi2f * SIN(pi2f * t2)
V2 = Q - e + c * temp
t2 = arcos((V2 + Vrec) / Vpeak) / pi2f

REM calculate convergence error
Vrip = V2 - V1
Error = ABS(Vrip - Vprev)
Vprev = Vrip
Vop = (V2 + V1) / 2
PRINT n; Error; Vop; Vrip

LOOP: REM go back and do it again

REM estimate peak ripple and rectifier currents
Iop = Ic + Vop / Rload
Irec = (Vpeak - Vrec - Vop) / Rsou
Irip = Irec - Iop
Inrush = (Vpeak - Vrec) / Rsou
```

```
REM list data values
PRINT : PRINT "constant current load="; Ic; "A"
PRINT "resistive load="; Rload; "ohms"
PRINT "supply voltage="; Vsup; "VRMS at"; freq;
"Hz"
PRINT "peak secondary voltage="; Vpeak; "V"
PRINT "effective source resistance="; Rsou;
"ohms"
PRINT "reservoir capacitor="; cap * 1000000!;
"microfarads"

REM list results
PRINT : PRINT "mean output="; Vop; "V, current=";
Iop; "A"
PRINT "ripple="; Vrip; "VPTP"
PRINT "ripple crest="; V2; "V, trough="; V1; "V"
PRINT "peak capacitor ripple current="; Irip; "A"
PRINT "peak repetitive rectifier current="; Irec;
"A"
PRINT "peak non-repetitive (inrush) current=";
Inrush; "A"
PRINT "approx inrush duration="; cr * 1000!;
"milliseconds"
PRINT "psu figure of merit="; pi2f * cap * Vop /
Iop
END

FUNCTION arcos (A)
REM no arc-cosine function provided in MS basic
arcos = ATN(SQR(1 / A / A - 1))
END FUNCTION
```